
C     *************************************************************** 
C 
C     ASCE Standardized Reference ET Equation for AgriMet 
C     January and February, 2011.  Peter L. Palmer 
C     AgriMet Program Manager, US Bureau of Reclamation, Boise, Idaho 
C     and Karl Tarbet, Hydromet Program Manager. 
C 
C     *************************************************************** 
C 
C     Program development from "Penman-Monteith daily (24 hour) 
C     Reference Evapotranspiration Equations for Estimating ETo, ETr, 
C     and HS ETo with Daily Data" by Snyder, R.L., University of 
C     California Davis, and Eching, S., California Department of 
C     Water Resources.  January 6, 2002, Revised February 2007. 
C     This document describes the implementation of the "ASCE 
C     Standardized Reference Evapotranspiration Equation", 
C     (ASCE-EWRI, 2004). 
C 
C     Note: Variable abbreviations in parentheses in the comments 
C           below are those used in the reference above. 
C 
C    ************************************************************** 
C                         List of Variables 
C    ************************************************************** 
C 
C 
C    AgriMet Archive Weather Parameter Variables: 
C    ------------------------------------------- 
C 
C    Tmaxf     -  Maximum Daily Air Temperature (Fahrenheit) 
C    Tminf     -  Minimum Daily Air Temperature (Fahrenheit) 
C    Tavgf     -  Average Daily Air Temperature (Fahrenheit) 
C    Tdewf     -  Mean Dew Point Temperature (Fahrenheit) 
C    Rlangley  -  Observed Solar Radiation  (Langleys/day) 
C    WRun      -  Total wind run at 2 meters (miles/day) 
C 
C 
C    AgriMet Station Variables: 
C    ------------------------- 
C 
C    Station   -  Four letter station identifier. 
C    Elev      -  Station Elevation in meters. 
C    Latdeg    -  Station Latitude in degrees (positive for North). 
C 
C 
C    Converted AgriMet Weather Parameter Variables: 
C    --------------------------------------------- 
C 
C    Tmax   (Tx)     -  Maximum Daily Air Temperature (Centigrade). 
C    Tmin   (Tn)     -  Minimum Daily Air Temperature (Centigrade). 
C    Tavg   (Tm)     -  Mean Daily Air Temperature computed as the mean 
C                       of the max and min daily air temperatures (C). 
C    Tdew   (Td)     -  Mean Daily Dew Point Temperature (Centigrade). 
C    WSavg  (U2)     -  Mean Daily Wind Speed at 2 meters, m/s. 
C    Rs     (Rs)     -  Measured solar radiation (MJ/m2/day) 
C 
C 



C    Humdity/Vapor Pressure Variables: 
C    --------------------------------- 
C 
C    LHvap   (lamda) -  Latent heat of vaporization: 2.45 MJ/kg 
C                       Inverse: 0.408 kg/MJ 
C    AtmPress (beta) -  Mean atmospheric (barometric) pressure at station 
C                       Elevation. Defined by a simple formulation of the 
C                       Universal Gas Law. 
C    PsyCon  (gamma) -  Psychrometric Constant.  Propotional to the mean 
C                       atmospheric pressure: PsyCon=0.000665*AtmPress 
C    SVP_Tmax (esTx) -  Saturation Vapor Pressure at the maximum daily 
C                       air temperature (Tmax). 
C    SVP_Tmin (esTn) -  Saturation Vapor Pressure at the minimum daily 
C                       air temperature (Tmin). 
C    SVP_Tavg (eo)   -  Saturation Vapor Pressure at the mean daily 
C                       air temperature (Tavg). 
C    SVP_Tdew (ea)   -  Actual vapor pressure or Saturation Vapor Pressure 
C                        at the daily mean dew point temperature (Tdew). 
C    SVPslope (delta)-  Slope of the Saturation Vapor Pressure-Temperature 
C                       Curve at mean air temperature (Tavg). 
C    SVPavg   (es)   -  Mean daily Saturation Vapor Pressure 
C    Ar       (Ar)   -  Aerodynamic term for tall canopy reference ET 
C    Ao       (Ao)   -  Aerodynamic term for short canopy reference ET 
C 
C 
C    Solar Radiation Variables: 
C    ------------------------- 
C 
C    Ra       (Ra)    -  Extraterrestrial Radiation - MJ/m2/day 
C    Rs       (Rs)    -  Incoming Solar Radiation - MJ/m2/day 
C    Rns      (Rns    -  Net Short Wave Radiation - MJ/m2/day 
C    Rnl      (Rnl)   -  Net Long Wave Radiation  - MJ/m2/day 
C    Rn       (Rn)    -  Net radiation over grass  - MJ/m2/day 
C    Rso      (Rso)   -  Clear Sky Solar Radiation (MJ/m2/day) (Calculated) 
C    JDay     (i)     -  Julian Day of the Year (1-366) 
C    Gsc      (Gsc)   -  Solar constant = 0.082 MJ/m/minute 
C    SBconst  (Sigma) -  Steffan-Boltzman constant = 4.90 x 10e-9 
C    Latrad   (Phi)   -  Latitude in radians = (PI*Latdeg)/180 
C    EccCorr  (dr)    -  Correction for eccentricity of Earth's orbit around 
C                        the sun on julian day of the year: 
C                        = 1+.033cos(2PI/365*Jday) 
C    SunDec (sm Delta)-  Declination of the sun above the celestial equator 
C                        in radians on julian day of the year: 
C                        = 0.409sin(2PI/365*Jday-1.39) 
C    PI               -  = 3.14159265358979323846 
C    SunAngle (Omega) -  Sunrise hour angle in radians 
C                        = cos e-1  (-tan Latrad x tan Sundec) 
C    Cloudf    (f)    -  A cloudiness function of Rs and Rso 
C                        = 1.35(Rs/Rso)-0.35 
C    Emis_sky  (eps') -  Apparent net clear sky emissivity 
C    Radr      (Rr)   -  Radiation term of ETrs equation for tall canopy. 
C    Rado      (Ar)   -  Radiation term of ETos equation for short canopy. 
C 
C 
 
 
 



C    Date Variables: 
C    -------------- 
C 
C    Month            - Month of the Year (1-12) 
C    Day              - Day of the Month (1-31) 
C    Year             - Year (four digits) 
C    JDay             - Julian day of the year 
C 
C 
C    ET Variables: 
C    ------------ 
C 
C    ETrs             - ASCE Standardized ET (Tall Crop Reference) 
C    ETos             - ASCE Standardized ET (Short Crop Reference) 
C 
C 
        subroutine ETS(*,Station,Month,Day,Year,Tmaxf,Tminf,Tavgf, 
     1                Rlangley,WRun,Tdewf,Etrs,ETos) 
        Integer Month,Day,Year 
        character*12 Station,STN 
        Real Elev,Latdeg,Tmaxf,Tminf,Tavgf,Rlangley,WRun,Tdewf 
        Real*4 CLDYC(5) 
  
        INTEGER JDay 
        REAL Latrad 
        REAL Lhvap 
        REAL*8 SBconst 
  
        INTEGER*4 IEND(13) 
C 
C    *************************************************************** 
C           Read in AgriMet Weather Station Location Parameters 
C           (Data is found in huser2:[agrimet.et]climat.dat) 
C 
C    Elev (El) -  Station Elevation in meters. 
C    Latdeg    -  Station Latitude in degrees (positive for North). 
C    Stn       -  Station Identifier read from climat.dat 
C 
        OPEN(UNIT=76,FILE='HUSER2:[AGRIMET.ET]CLIMAT.DAT',TYPE='OLD', 
     1  READONLY,IOSTAT=IOS,recl=86) 
        IF(IOS.NE.0) THEN 
                WRITE(*,*) '  ERROR OPENING CLIMAT.DAT etos.for:160' 
                GO TO 950 
        END IF 
 30      READ(76,'(A5,3F9.0,F10.0,F9.0,F6.0,F4.0,2F7.0,F5.0,F6.0)', 
     2  END=1000, 
     3  IOSTAT=ISTAT)STN,CLDYC,CLDYM,ANEMH,ELEV,CSUBT,TSUBX,Latdeg 
        IF(ISTAT.NE.0) THEN 
                WRITE(*,*) '  ERROR READING CLIMAT.DAT' 
                GO TO 925 
        END IF 
        IF(STN.NE.STATION)GO TO 30 
  
         Write(*,*)'Etos:  Elev=',Elev,' LatDeg= ',LatDeg 
C 
C 
 



C    *************************************************************** 
C 
C       Convert AgriMet Tmaxf, Tminf, Tavgf, and Tdewf to degrees C 
C 
C       Convert AgriMet wind run(Wrun miles/day)to average wind speed m/s. 
C       meters per second  = miles per day/24/2.236936 
C 
C    *************************************************************** 
C 
        Tmax=5.*(Tmaxf-32.)/9. 
        Tmin=5.*(Tminf-32.)/9. 
        Tavg=5.*(Tavgf-32.)/9. 
        Tdew=5.*(Tdewf-32.)/9. 
        WSavg=Wrun/24./2.236936 
C 
C 
C    *************************************************************** 
C       Compute Julian Day of the Year (JDay) (1-366) for 
C       subroutine call input date.  Procedure from Appendix A, 
C       Table A5, equation A5.1, ASCE Manual 70 Revision, 2011. 
C 
C    *************************************************************** 
C 
        Jday=Day-32+INT(275*Month/9)+2*INT(3/(Month+1))+ 
     1  INT(Month/100-(MOD(Year,4)/4)+0.975) 
C 
C    *************************************************************** 
C       Limit AgriMet solar radiation (langleys) to less than 800 to 
C       prevent extreme ET values due to rollover or bad data. 
C 
C       Convert AgriMet Solar Radiation(langleys) to MJ/m2/day 
C       Note that AgriMet Langleys are computed using the 
C       Meteorology 15C Calorie, defined as 4.1855 J/cm2. 
C       Therefore, Rs(MJ/m2/day) = .041855 * Rlangley 
C       Note that ASCE FAO 56 uses ITcal, defined as 
C       4.1868 joules, but we need to use the same conversion here 
C       as implemented in the AgriMet database for consistency. 
C       The difference is only 0.03%, 100 times less than the 
C       accuracy of the Licor pyranometer. 
C 
C    *************************************************************** 
C 
C 
        If(Rlangley.GT.900.0)Rlangley=800.0 
        Rs=Rlangley*0.041855 
C 
C 
C    *************************************************************** 
C       Compute Extraterrestrial Radiation (Ra) using equations 
C       From Duffie and Beckman (1980), "Solar engineering of thermal 
C       processes". John Wiley and Sons, New York, pp. 1-109. 
C 
C       Ra= (24*60/PI) * Gsc * EccCorr * (SunAngle * sin Sundec 
C            *sin LatRad + cos Latrad *cos Sundec *sin Sunangle) 
C 
 
 



C       Where: 
C 
C       Gsc     = Solar constant = 0.082 MJ/m2/minute 
C       EccCorr = Correction for eccentricity of Earth's orbit around 
C                 the sun on julian day of the year: 
C                 = 1+.033cos(2PI/365*Jday) 
C       Latrad   = Latitude in radians = (PI*Latdeg)/180 
C       SunDec   = Declination of the sun above the celestial equator 
C                  in radians on julian day of the year: 
C                  = 0.409sin(2PI/365*Jday-1.39) 
C       SunAngle = Sunrise hour angle in radians 
C                  = arccos(-tan Latrad x tan Sundec) 
C 
C    *************************************************************** 
C 
        PI=3.14159265358979323846 
        Gsc=0.082 
        EccCor=1+(0.033*cos(2*PI/365*Jday)) 
        Latrad=(PI*Latdeg)/180 
        SunDec=0.409*sin(2*PI/365*Jday-1.39) 
        SunAngle=(acos(-tan(Latrad)*tan(Sundec))) 
        Ra= 24*60/PI*Gsc*EccCor*(SunAngle*sin(Sundec)* 
     1      sin(LatRad)+cos(Latrad)*cos(Sundec)*sin(Sunangle)) 
C 
C    *************************************************************** 
C       Compute Net Radiation (Rn) in MJ/m/day using equations 
C       from Allen et al. (1994). "An update for the calculation 
C       of reference evapotranspiration".  ICIC Bulletin 1994 
C       Volume 43 No 2. 
C 
C       Rso      (Rso)   = Clear sky total global solar radiation at 
C                            the earth's surface in MJ/m2/day 
C                            = Ra(0.75 + 2.0 x 10e-5  * Elev) 
C       Rns      (Rns)   = Net solar radiation over grass as function of 
C                            measured solar radiation in MJ/m2/day 
C                            = (1-0.23)Rs 
C       Cloudf   (f)     = Cloudiness function = 1.35 * Rs/Rs0 - 0.35 
C       SVP_Tmax (esTx)  = Saturation Vapor Pressure (kPa) at the 
C                            maximum daily air temperature in Centigrade: 
C                            = 0.6108 exp (17.27 * Tmax / Tmax + 237.3) 
C       SVP_Tmin (esTn)  = Saturation Vapor Pressure (kPa) at the minimum 
C                            daily air temperature in Centigrade: 
C                            = 0.6108 exp (17.27 * Tmin / Tmin + 237.3) 
C       SVP_Tavg (eo)    = Saturation Vapor Pressure (kPa) at the mean 
C                            daily air temperature in Centigrade: 
C                            = 0.6108 exp (17.27 * Tavg / Tavg + 237.3) 
C       SVP_Tdew  (ea)   = Actual vapor pressure or saturation vapor pressure 
C                            at the daily mean dew point (Tdew) temperature: 
C                            = 0.6108exp(17.27Tdew/Tdew+237.3) 
C       SVPavg    (es)   = Mean daily Saturation Vapor Pressure: 
C                            = (SVP_Tmax + SVP_Tmin) / 2 
C       Emis_sky   (e')  = Apparent net clear sky emissivity 
C                            = 0.34 - 0.14 sqrt(SVP_Tdew) 
C       SBconst    (o)   = Steffan-Boltzman constant = 4.90 x 10e-9 
C       Rnl        (Rnl) = Net long wave radiation in MJ/m2/day 
C       Rn         (Rn)  = Net radiation over grass in MJ/m2/day 
C                            = Rns + Rnl 



C       Radr       (Rr)  = Radiation term of ETrs equation for tall canopy. 
C       Rado       (Ro)  = Radiation term of ETos equation for short canopy. 
C 
C    *************************************************************** 
C 
        Rso= Ra*(0.75+0.00002*Elev) 
        Rns=(1-0.23)*Rs 
        Cloudf=(1.35*Rs/Rso)-0.35 
        SVP_Tmax=0.6108*exp(17.27*Tmax/(Tmax+237.3)) 
        SVP_Tmin=0.6108*exp(17.27*Tmin/(Tmin+237.3)) 
        SVP_Tavg=0.6108*exp(17.27*Tavg/(Tavg+237.3)) 
        SVP_Tdew=0.6108*exp(17.27*Tdew/(Tdew+237.3)) 
        SVPavg=(SVP_Tmax+SVP_Tmin)/2 
        Emis_sky =0.34-(0.14*sqrt(SVP_Tdew)) 
        SBconst=4.90/10**9 
        Rnl=-
Cloudf*Emis_sky*Sbconst*(((Tmax+273.15))**4+((Tmin+273.15)**4))/2 
        Rn=Rns+Rnl 
C 
C    *************************************************************** 
C       Compute Mean Atmospheric Pressure as a function of station 
C       elevation in meters (Elev).  AtmPress is a simplified 
C       formulation of the Universal Gas Law. 
C 
C                                             5.26 
C                            (293-0.0065 Elev) 
C       AtmPress (B) = 101.3 (---------------) 
C                            (      293      ) 
C 
C 
C    *************************************************************** 
C 
        AtmPress = 101.3 * ((((293-0.0065*Elev)/293))**5.26) 
C 
C    *************************************************************** 
C       Compute Psychometric Constant (PsyCon) 
C       Proportional to the mean atmospheric pressure and latent heat 
C       Of vaporization (2.45 MJ/kg). 
C 
C       Lhvap  (lambda) = 2.45 MJ/kg 
C       PsyCon (gamma)  = 0.00163(AtmPress/LHvap) 
C    *************************************************************** 
C 
        Lhvap=2.45 
        PsyCon=0.00163*(AtmPress/LHvap) 
C 
C 
C    *************************************************************** 
C       Compute the slope of the Saturation Vapor Pressure Curve 
C       at the mean air temperature (Tavg): 
C 
C       SVPSlope =      4099 * SVP_Tavg 
C                    ------------------- 
C                                    2 
C                      (Tavg + 237.3) 
C 
 



C    *************************************************************** 
C 
        SVPSlope = (4099*SVP_Tavg)/((Tavg+237.2)**2) 
C 
C    *************************************************************** 
C 
C       Compute ETrs using the ASCE-EWRI (2004) standardized equation 
C       for tall canopy (alfalfa) reference ET. 
C 
C       Radr = radiation term of the equation for tall canopy. 
C       G    = soil heat flux density in MJ/m2/day = 0 
C 
C                  0.408 SVPslope (Rn - G) 
C       Radr = ------------------------------- 
C              SVPslope + PsyCon(1+0.38 WSavg) 
C 
C       Ar   = Aerodynamic term of the equation for tall canopy. 
C 
C            (1600 PsyCon) 
C            (-----------) WSavg(SVPavg-SVP_Tdew) 
C            ( Tavg +273 ) 
C       Ar = (----------------------------------- 
C                SVPslope + PsyCon (1+0.38 WSavg) 
C 
C       ETrs = Radr + Ar 
C 
C 
        G=0.00 
        Radr=(0.408*SVPslope*(Rn - G))/(SVPslope+PsyCon*(1+0.38*WSavg)) 
        Ar=((1600*PsyCon)/(Tavg+273))*WSavg*(SVPavg-SVP_Tdew)/ 
     1     (SVPslope+PsyCon*(1+0.38*WSavg)) 
        ETrs=Radr+Ar 
C 
C 
C    *************************************************************** 
C 
C       Compute ETos using the ASCE-EWRI (2004) standardized equation 
C       for short canopy (grass) reference ET. 
C 
C       Rado = radiation term of the equation for short canopy. 
C 
C       G    = soil heat flux density in MJ/m2/day = 0 
C 
C                  0.408 SVPslope (Rn - G) 
C       Rado = ------------------------------- 
C              SVPslope + PsyCon(1+0.34 WSavg) 
C 
C       Ao   = Aerodynamic term of the equation for tall canopy. 
C 
C            (900  PsyCon) 
C            (-----------) WSavg(SVPavg-SVP_Tdew) 
C            ( Tavg +273 ) 
C       Ao = (----------------------------------- 
C                SVPslope + PsyCon (1+0.34 WSavg) 
C 
C       ETro = Rado + Ao 
C 



C 
        G=0.00 
        Rado=(0.408*SVPslope*(Rn - G))/(SVPslope+PsyCon*(1+0.34*WSavg)) 
        Ao=((900*PsyCon)/(Tavg+273))*WSavg*(SVPavg-SVP_Tdew)/ 
     1     (SVPslope+PsyCon*(1+0.34*WSavg)) 
        ETos=Rado+Ao 
C 
C 
C    ************************************************************** 
C              Convert ETrs and ETos from mm to inches 
C    ************************************************************** 
C 
       ETos=ETos/25.4 
       ETrs=ETrs/25.4 
C 
C    *************************************************************** 
C                  Printout Variable Names and Values 
C    *************************************************************** 
C 
        PRINT *,'Jday',JDay 
        PRINT *,'Elev',Elev 
        PRINT *,'Latdeg',Latdeg 
        PRINT *,'Latrad',Latrad 
        PRINT *,'Tmaxf',Tmaxf 
        PRINT *,'Tminf',Tminf 
        PRINT *,'Tavgf',Tavgf 
        PRINT *,'Tdewf',Tdewf 
        PRINT *,'Rlangley',Rlangley 
        PRINT *,'WRun',WRun 
        PRINT *,'Tmax (Tx)',Tmax 
        PRINT *,'Tmin (Tn)',Tmin 
        PRINT *,'Tavg (Tm)',Tavg 
        PRINT *,'Tdew (Td)',Tdew 
        PRINT *,'WSavg (U2)',WSavg 
        PRINT *,'PI',PI 
        PRINT *,'Gsc',Gsc 
        PRINT *,'EccCor (dr)',EccCor 
        PRINT *,'SunDec (o^)',SunDec 
        PRINT *,'SunAngle (Ws)',SunAngle 
        PRINT *,'Cloudf (f)',Cloudf 
        PRINT *,'Emis_sky (e)',Emis_sky 
        PRINT *,'SBconst (sigma)',SBconst 
        PRINT *,' ' 
        PRINT *,'Rs',Rs 
        PRINT *,'Ra',Ra 
        PRINT *,'Rso',Rso 
        PRINT *,'Rns',Rns 
        PRINT *,'Rn',Rn 
        Print *,'Rnl',Rnl 
        PRINT *,' ' 
        PRINT *,'SVP_Tmax (EsTx)',SVP_Tmax 
        PRINT *,'SVP_Tmin (EsTn)',SVP_Tmin 
        PRINT *,'SVP_Tavg (Eo)',SVP_Tavg 
        PRINT *,'SVP_Tdew (Ea)',SVP_Tdew 
        PRINT *,'SVPAvg   (Es)',SVPavg 
        PRINT *,'SVPslope (delta)',SVPslope 
        PRINT *,'AtmPress (beta)',AtmPress 



        PRINT *,'Lhvap (lamda)',Lhvap 
        PRINT *,'PsyCon (gamma)',PsyCon 
        PRINT *,'G',G 
        PRINT *,'Radr for ETrs (Rr)',Radr 
        PRINT *,'Ar for ETrs (Ar)',Ar 
        PRINT *,'Rado for ETos (Ro)',Rado 
        PRINT *,'Ao for ETos (Ao)',Ao 
        PRINT *,'ETrs',ETrs 
        PRINT *,'ETos',ETos 
C 
        CLOSE(UNIT=76,ERR=1002) 
C 
        RETURN 
C 
1000    WRITE(6,1001,ERR=925) STATION 
1001    FORMAT(' %ETPCOMP-1001-STATION NOT FOUND: ',A) 
925     CLOSE(UNIT=76,ERR=950) 
950     RETURN 1 
C 
1002    RETURN 
        END 


